Numerikus Matematika — 12
Gyakorló Feladatok

Hermite Interpoláció

Probléma:

Adott:

x={x1,..., xr}
y = {y1$^{(0)}$, ..., yr$^{(0)}$}
yd = {y1$^{(1)}$, ..., yr$^{(1)}$}
(...)
vagy (általánosabban)

x={x1,..., xr}
{y1$^{(0)}$, .., y1$^{(m1-1)}$}
...
{yr$^{(0)}$, ..., yr$^{(mr-1)}$}

Keressünk ((m1+...+mr)−1)−edfokú polinomot melyre $d^j\ f / d\, x^j [x_k] = y_k^{(j)}$

▪ Példa

Adjunk olyan polinomot, ami másodrenben érinti 0−ban és π−ben a sin fgv−t!

x= {0,π}
y= {0,0}
yd= {1,−1}
ydd={0,0}

m1=3,m2=3 $\Longrightarrow$legf. ötödfokú pol.

Alternatív leirás {0,π}, {0,1,0}, {0,−1,0}
Elôny: lehetnek a listák kül. hosszúak.

Mathematica: beépített parancs InterpolatingPolynomial[]

```
In[1]:=
        DT = Table[D[Sin[x], {x, j}], {j, 0, 2}]

Out[1]= {Sin[x], Cos[x], -Sin[x]}

In[2]:= DT /. x → 0

Out[2]= {0, 1, 0}
```

```
In[3]:=  DT /. x → π

Out[3]=  {0, -1, 0}

In[4]:=  p = InterpolatingPolynomial[{{0, {0, 1, 0}}, {π, {0, -1, 0}}}, x] // Expand
```

Out[4]= $x - \frac{2 x^3}{\pi^2} + \frac{x^4}{\pi^3}$

A feltételek ellenôrzése:

```
In[5]:=  p /. {{x → 0}, {x → π}}

Out[5]=  {0, 0}

In[6]:=  D[p, x] /. {{x → 0}, {x → π}}

Out[6]=  {1, -1}

In[7]:=  D[p, x, x] /. {{x → 0}, {x → π}}

Out[7]=  {0, 0}
```

Most az alg szerint megkonstruáljuk p−−t!

```
In[8]:=  p0 = Sin[x] /. x → 0

Out[8]=  0

In[9]:=  p1 = p0 + A (x - 0)

Out[9]=  A x

In[10]:=  Solve[(D[p1, x] /. x → 0) == (D[Sin[x], x] /. x → 0), A]

Out[10]=  {{A → 1}}

In[11]:=  p1 = p0 + 1 (x - 0)

Out[11]=  x

In[12]:=  p2 = p1 + A (x - 0) (x - 0)
```

Out[12]= $x + A x^2$

```
In[13]:=  Solve[(D[p2, x, x] /. x → 0) == (D[Sin[x], x, x] /. x → 0), A]

Out[13]=  {{A → 0}}

In[14]:=  p2 = p1 + 0 (x - 0) (x - 0)

Out[14]=  x
```

Vegyük észre hgy ez egy Taylor polinom:

```
In[15]:= Series[Sin[x], {x, 0, 2}] // Normal
```

Out[15]= x

```
In[16]:= p3 = p2 + A (x - 0) (x - 0) (x - 0)
```

Out[16]= $x + A x^3$

```
In[17]:= Solve[(p3 /. x → π) == (Sin[x] /. x → π), A]
```

Out[17]= $\left\{\left\{A \to -\frac{1}{\pi^2}\right\}\right\}$

```
In[18]:= p3 = p3 /. Solve[(p3 /. x → π) == (Sin[x] /. x → π), A][[1]]
```

Out[18]= $x - \frac{x^3}{\pi^2}$

```
In[19]:= p4 = p3 + A (x - 0) (x - 0) (x - 0) (x - π)
```

Out[19]= $x - \frac{x^3}{\pi^2} + A x^3 (-\pi + x)$

```
In[20]:= Solve[(D[p4, x] /. x → π) == (D[Sin[x], x] /. x → π), A]
```

Out[20]= $\left\{\left\{A \to \frac{1}{\pi^3}\right\}\right\}$

```
In[21]:= p4 = p4 /. Solve[(D[p4, x] /. x → π) == (D[Sin[x], x] /. x → π), A][[1]]
```

Out[21]= $x - \frac{x^3}{\pi^2} + \frac{x^3 (-\pi + x)}{\pi^3}$

```
In[22]:= p5 = p4 + A (x - 0) (x - 0) (x - 0) (x - π) (x - π)
```

Out[22]= $x - \frac{x^3}{\pi^2} + \frac{x^3 (-\pi + x)}{\pi^3} + A x^3 (-\pi + x)^2$

```
In[23]:= Solve[(D[p5, x, x] /. x → π) == (D[Sin[x], x, x] /. x → π), A]
```

Out[23]= $\{\{A \to 0\}\}$

```
In[24]:= p5 = p5 /. Solve[(D[p5, x, x] /. x → π) == (D[Sin[x], x, x] /. x → π), A][[1]]
```

Out[24]= $x - \frac{x^3}{\pi^2} + \frac{x^3 (-\pi + x)}{\pi^3}$

```
In[25]:= p = Expand[p5]
```

Out[25]= $x - \frac{2 x^3}{\pi^2} + \frac{x^4}{\pi^3}$

```
In[26]:= InterpolatingPolynomial[{{0, {0, 1, 0}}, {π, {0, -1, 0}}}, x] // Expand
```

Out[26]= $x - \frac{2 x^3}{\pi^2} + \frac{x^4}{\pi^3}$

```
In[27]:= Plot[{Sin[x], p}, {x, -.2, Pi + .2},
           PlotStyle -> {{RGBColor[0, 0, 1]}, {RGBColor[1, 0, 0]}}];
```



Második megoldás

In[28]:=

$$p = \sum_{j=0}^{5} \alpha_j x^j$$

Out[28]= $\alpha_0 + x\,\alpha_1 + x^2\,\alpha_2 + x^3\,\alpha_3 + x^4\,\alpha_4 + x^5\,\alpha_5$

```
In[29]:= Join[Table[D[Sin[x], {x, j}] == D[p, {x, j}], {j, 0, 2}] /. x → 0,
          Table[D[Sin[x], {x, j}] == D[p, {x, j}], {j, 0, 2}] /. x → π]
```

Out[29]= $\{0 == \alpha_0,\ 1 == \alpha_1,\ 0 == 2\,\alpha_2,\ 0 == \alpha_0 + \pi\,\alpha_1 + \pi^2\,\alpha_2 + \pi^3\,\alpha_3 + \pi^4\,\alpha_4 + \pi^5\,\alpha_5,$
$-1 == \alpha_1 + 2\,\pi\,\alpha_2 + 3\,\pi^2\,\alpha_3 + 4\,\pi^3\,\alpha_4 + 5\,\pi^4\,\alpha_5,\ 0 == 2\,\alpha_2 + 6\,\pi\,\alpha_3 + 12\,\pi^2\,\alpha_4 + 20\,\pi^3\,\alpha_5\}$

```
In[30]:= Solve[%]
```

Out[30]= $\left\{\left\{\alpha_0 \to 0,\ \alpha_3 \to -\frac{2}{\pi^2},\ \alpha_4 \to \frac{1}{\pi^3},\ \alpha_5 \to 0,\ \alpha_1 \to 1,\ \alpha_2 \to 0\right\}\right\}$

```
In[31]:= p /. %
```

Out[31]= $\left\{x - \frac{2\,x^3}{\pi^2} + \frac{x^4}{\pi^3}\right\}$

▪ 2. Példa

x={x1,x2,x3}={1,2,3}
y={y1,y2,y3}={2,32,242}
yd={yd1,yd2,yd3}={4,79,404}

Adjuk meg a Hermite interpolációs polinomot!

```
In[79]:= p1 = 2 + 4 (x - 1) // Expand
```

Out[79]= $-2 + 4\,x$

```
p2 = p1 + A (x - 1) (x - 1)

Out[81]=  -2 + A (-1 + x)^2 + 4 x

In[82]:=  Solve[(p2 /. x → 2) == 32, A]

Out[82]=  {{A → 26}}

In[83]:=  p2 = p1 + 26 (x - 1)^2

Out[83]=  -2 + 26 (-1 + x)^2 + 4 x

In[69]:=  p3 = p2 + A (x - 1)^2 (x - 2)

Out[69]=  -2 + 26 (-1 + x)^2 + A (-2 + x) (-1 + x)^2 + 4 x

In[70]:=  Solve[(D[p3, x] /. x → 2) == 79, A]

Out[70]=  {{A → 23}}

In[84]:=  p3 = p2 + 23 (x - 1)^2 (x - 2)

Out[84]=  -2 + 26 (-1 + x)^2 + 23 (-2 + x) (-1 + x)^2 + 4 x

In[93]:=  p4 = p3 + A (x - 1)^2 (x - 2) (x - 2)

Out[93]=  -2 + 26 (-1 + x)^2 + 23 (-2 + x) (-1 + x)^2 + A (-2 + x)^2 (-1 + x)^2 + 4 x

In[95]:=  Solve[(p4 /. x → 3) == 242, A]

Out[95]=  {{A → 9}}

In[96]:=  p4 = p3 + 9 (x - 1)^2 (x - 2) (x - 2)

Out[96]=  -2 + 26 (-1 + x)^2 + 23 (-2 + x) (-1 + x)^2 + 9 (-2 + x)^2 (-1 + x)^2 + 4 x

In[97]:=  p5 = p4 + A (x - 1)^2 (x - 2)^2 (x - 3)

Out[97]=  -2 + 26 (-1 + x)^2 + 23 (-2 + x) (-1 + x)^2 + 9 (-2 + x)^2 (-1 + x)^2 + A (-3 + x) (-2 + x)^2 (-1 + x)^2 + 4 x

In[98]:=  Solve[(D[p5, x] /. x → 3) == 404, A]

Out[98]=  {{A → 1}}

In[101]:=
    p5 = p4 + 1 (x - 1)^2 (x - 2)^2 (x - 3)

Out[101]=
    -2 + 26 (-1 + x)^2 + 23 (-2 + x) (-1 + x)^2 + 9 (-2 + x)^2 (-1 + x)^2 + (-3 + x) (-2 + x)^2 (-1 + x)^2 + 4 x

In[102]:=
    Expand[%]

Out[102]=
    2 - x + x^5
```

Ellenôrzés

```
In[32]:= InterpolatingPolynomial[{{1, {2, 4}}, {2, {32, 79}}, {3, {242, 404}}}, x] // Expand
```

Out[32]= $2 - x + x^5$

Legkisebb négyzetek módszere

■ Kalkulus, normálegyenletek

M=1

x={1,2,3}, y={2,5,10}

```
In[35]:= F0[x_] := α0;

In[36]:=
X = {1, 2, 3};
Y = {2, 5, 10};

In[39]:= G = Plus @@ Map[(F0[#[[1]]] - #[[2]])^2 &, Transpose[{X, Y}]]
```

Out[39]= $(-10 + \alpha0)^2 + (-5 + \alpha0)^2 + (-2 + \alpha0)^2$

```
In[40]:= D[G, α0]
```

Out[40]= $2 (-10 + \alpha0) + 2 (-5 + \alpha0) + 2 (-2 + \alpha0)$

```
In[41]:= Solve[D[G, α0] == 0]
```

Out[41]= $\{\{\alpha0 \to \frac{17}{3}\}\}$

```
In[42]:= Graphics[{RGBColor[1, 0, 0], PointSize[.03], Point[{1, 2}], Point[{2, 5}],
   Point[{3, 10}], RGBColor[0, 1, 0], Line[{{0, 17 / 3}, {4, 17 / 3}}]}];

In[43]:= Show[%, PlotRange → {1, 11}];
```



M=2

```
In[44]:= F1[x_] := α1 x + α0;
```

A négyzetösszeg: ⟶ G=G[α0,α1] kétváltozós fgv.

```
In[45]:=
         G = Plus @@ Map[(F1[#[[1]]] - #[[2]])^2 &, Transpose[{X, Y}]]

Out[45]= (-2 + α0 + α1)^2 + (-5 + α0 + 2 α1)^2 + (-10 + α0 + 3 α1)^2
```

Szüks felt., normálegyenelet.

```
In[46]:= {D[G, α0], D[G, α1]}

Out[46]= {2 (-2 + α0 + α1) + 2 (-5 + α0 + 2 α1) + 2 (-10 + α0 + 3 α1),
          2 (-2 + α0 + α1) + 4 (-5 + α0 + 2 α1) + 6 (-10 + α0 + 3 α1)}

In[47]:= Solve[{D[G, α0] == 0, D[G, α1] == 0}]

Out[47]= {{α0 → -7/3, α1 → 4}}
```

A kapott lin. polinom a 4x−7/3!

```
In[48]:= Graphics[{RGBColor[1, 0, 0], PointSize[.03], Point[{1, 2}], Point[{2, 5}],
            Point[{3, 10}], RGBColor[0, 1, 0], Line[{{0, -7/3}, {4, 41/3}}]}];

In[49]:= Show[%, PlotRange → {1, 11}];
```



M=3

```
In[50]:= F2[x_] := α2 x^2 + α1 x + α0;

In[51]:=
         G = Plus @@ Map[(F2[#[[1]]] - #[[2]])^2 &, Transpose[{X, Y}]]

Out[51]= (-2 + α0 + α1 + α2)^2 + (-5 + α0 + 2 α1 + 4 α2)^2 + (-10 + α0 + 3 α1 + 9 α2)^2

In[52]:= {D[G, α0], D[G, α1], D[G, α2]}

Out[52]= {2 (-2 + α0 + α1 + α2) + 2 (-5 + α0 + 2 α1 + 4 α2) + 2 (-10 + α0 + 3 α1 + 9 α2),
          2 (-2 + α0 + α1 + α2) + 4 (-5 + α0 + 2 α1 + 4 α2) + 6 (-10 + α0 + 3 α1 + 9 α2),
          2 (-2 + α0 + α1 + α2) + 8 (-5 + α0 + 2 α1 + 4 α2) + 18 (-10 + α0 + 3 α1 + 9 α2)}
```

```
In[53]:= Solve[{D[G, α0] == 0, D[G, α1] == 0, D[G, α2] == 0}]

Out[53]= {{α0 → 1, α1 → 0, α2 → 1}}
```

A kapott kvadratikus polinom az $x^2 + 1$!

```
In[54]:= InterpolatingPolynomial[Transpose[{X, Y}], x] // Expand

Out[54]= 1 + x^2

In[55]:=
  Plot[Evaluate[InterpolatingPolynomial[Transpose[{X, Y}], x]],
    {x, 0, 4}, ImageSize → {300, 300}, PlotStyle → {RGBColor[0, 1, 0]},
    Epilog → {RGBColor[1, 0, 0], PointSize[.03], Map[Point[#] &, Transpose[{X, Y}]]},
    PlotRange → {0, 20}];
```



A LS min. probléma felfogható a Lagrange interp. probléma általánosításaként!